# $Id: PKGBUILD 123175 2011-05-09 02:48:19Z eric $
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=fluxter
pkgver=0.1.0
pkgrel=5
pkgdesc="A workspace pager for fluxbox"
arch=('i686' 'x86_64')
url="http://benedict.isomedia.com/homes/stevencooper/projects/fluxter.html"
license=('GPL')
depends=('gcc-libs' 'libx11')
source=(http://benedict.isomedia.com/homes/stevencooper/files/${pkgname}-${pkgver}.tar.gz \
        fluxter-0.1.0-asneeded.patch)
md5sums=('6d18553220e8fc33c54762d2e7d31528'
         '3deb4e816d12a262455bc8281fa82577')
sha1sums=('d0da4759a21fdadc8a6457195c87b6648b5d69bc'
          '2fc912d40233577b34cb6c641e066863e487fab0')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  patch -p0 -i ../fluxter-0.1.0-asneeded.patch
  autoreconf
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
